
1 Overview

This document lists all changes needed to correct problems in the TTCN implementation of test case 8.5.4.4 which is part of the LTE/SAE test suite in the ‘iwd-TTCN3-B2015-03_D16wk24’ ATS delivery.

The test case can be demonstrated to run one LTE UE (see section 5). Execution logs are provided as evidence.. 

3 Verification Test Summary

Test Case:
TC_8_5_4_4
Test Group:
LTE_A_R12_IWD_16wk10\LTE_A_R12\RRC_Others_R12.ttcn
ATS Version:
iwd-TTCN3-B2015-03_D16wk24
System Simulator used:
Anite Conformance Toolset Solution

UE used:
Sequans Calliope-Cat0
Verification Status:
PASS

4 Corrections required for test case 8.5.4.4
Change 1 – Correction to Function “f_EUTRA_508_RRCConnectionSetup”
	Function name
	f_EUTRA_508_RRCConnectionSetup 

	Reason for change
	The current test model to test Low Cost MTC test cases in half duplex operation restricts UE UL transmission to subframe #2 only and DL reception to subframe #8 only. Because of this, enabling CQI and SRS reporting shall conflict with the above frame restrictions to be applied, which could result in device may not always be able to receive the DL messages sent on subframe #8, due to conflicts with CQI/SRS subframes. Hence it is proposed to disable CQI and SRS reporting if the test case is executed in halfduplex mode.

Note: A draft prose CR is associated with this change.

	Summary of change
	New Templates are defined for RRCConnectionSetup and RRCConnectionReconfiguration messages to disable CQI and SRS reporting. And these new templates are called instead of the default templates based on the capability of UE to support half duplex operation, pc_FDD_TypeB_HalfDuplex (Note) is used to check the same. New templates defined are listed below:

1.
cs_508_PhysicalConfigDedicated_Default_RBC_HO_TypeB_HalfDuplex

2.
cs_508_PhysicalConfigDedicated_Default_RBC_TypeB_HalfDuplex

3.
cds_508_PhysicalConfigDedicated_Default_SRB1_TypeB_HalfDuplex

4.
cs_SRB2_1AM_DRB_Config_Def_TypeB_HalfDuplex

5.
cs_508_RRCConnectionSetup_TypeB_HalfDuplex

6.
cs_RRCConnectionReconfiguration_SRB2_1AM_DRB_Est_TypeB_HalfDuplex

Note: Please note that this checking should be based on a pixit parameter. So probably, a prose CR is required for adding a nex pixit parameter for testing in Half Duplex operation and that should be used to enforce half duplex changes. We can provide a draft if it is agreed.

	TTCN module
	EUTRA_ConfigurationSteps.ttcn, EUTRA_NASSteps.ttcn, EUTRA_RRCSteps.ttcn

	MCC160 Comment
	


EUTRA_ConfigurationSteps.ttcn
Before change

  function f_EUTRA_508_RRCConnectionSetup(EUTRA_CellId_Type p_CellId,

                                          RRC_TransactionIdentifier p_RRC_TI := tsc_RRC_TI_Def,

                                          TimeAlignmentTimer p_TimeAlignmentTimer := sf750) runs on EUTRA_PTC return template (value) DL_CCCH_Message

  { /* @sic function introduced for implementation of R5-120721 change 3 sic@ */

    var EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_Info := f_EUTRA_CellInfo_GetFDD_TDD_Info(p_CellId);

    var SR_PUCCH_ResourceIndex_Type v_Sr_PUCCH_ResourceIndex := f_EUTRA_CellInfo_GetSR_PUCCH_ResourceIndex(p_CellId);

    var PhysicalConfigDedicated_AntennaInfo_Type v_AntennaInfo := f_EUTRA_CellInfo_GetAntennaInfoDedicated(p_CellId);

    var SR_ConfigIndex_Type v_SR_ConfigIndex := f_EUTRA_CellInfo_GetSR_ConfigIndex(p_CellId);

  return cs_508_RRCConnectionSetup(p_RRC_TI,

                                     v_EUTRA_FDD_TDD_Info.pucch_Config,

                                     v_EUTRA_FDD_TDD_Info.cqi_ReportConfig,        // @sic R5-120497 sic@

                                     v_EUTRA_FDD_TDD_Info.soundingRS_UL_Config,    // @sic R5-120497 sic@

                                     v_Sr_PUCCH_ResourceIndex,

                                     v_AntennaInfo,

                                     v_SR_ConfigIndex,

                                     p_TimeAlignmentTimer);
  }

After change

  function f_EUTRA_508_RRCConnectionSetup(EUTRA_CellId_Type p_CellId,

                                          RRC_TransactionIdentifier p_RRC_TI := tsc_RRC_TI_Def,

                                          TimeAlignmentTimer p_TimeAlignmentTimer := sf750) runs on EUTRA_PTC return template (value) DL_CCCH_Message

  { /* @sic function introduced for implementation of R5-120721 change 3 sic@ */

    var EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_Info := f_EUTRA_CellInfo_GetFDD_TDD_Info(p_CellId);

    var SR_PUCCH_ResourceIndex_Type v_Sr_PUCCH_ResourceIndex := f_EUTRA_CellInfo_GetSR_PUCCH_ResourceIndex(p_CellId);

    var PhysicalConfigDedicated_AntennaInfo_Type v_AntennaInfo := f_EUTRA_CellInfo_GetAntennaInfoDedicated(p_CellId);

    var SR_ConfigIndex_Type v_SR_ConfigIndex := f_EUTRA_CellInfo_GetSR_ConfigIndex(p_CellId);

    //Anite@LCMTC

    if (pc_FDD_TypeB_HalfDuplex)

    {

        return cs_508_RRCConnectionSetup_TypeB_HalfDuplex(p_RRC_TI,

                                     v_EUTRA_FDD_TDD_Info.pucch_Config, 

                                     v_Sr_PUCCH_ResourceIndex,

                                     v_AntennaInfo,

                                     v_SR_ConfigIndex,

                                     p_TimeAlignmentTimer);

    }

    else

    {
        return cs_508_RRCConnectionSetup(p_RRC_TI,

                                     v_EUTRA_FDD_TDD_Info.pucch_Config,

                                     v_EUTRA_FDD_TDD_Info.cqi_ReportConfig,        // @sic R5-120497 sic@

                                     v_EUTRA_FDD_TDD_Info.soundingRS_UL_Config,    // @sic R5-120497 sic@

                                     v_Sr_PUCCH_ResourceIndex,

                                     v_AntennaInfo,

                                     v_SR_ConfigIndex,

                                     p_TimeAlignmentTimer);

    }
  }

EUTRA_NASSteps.ttcn
Before change:
     function f_EUTRA_RRCConnectionReconfiguration_Common(EUTRA_CellId_Type p_CellId,

                                                       RRC_TransactionIdentifier p_RRC_TI,

                                                       UInt_Type p_NoOfAddDRBs_AM := 0,

                                                       UInt_Type p_NoOfAddDRBs_UM := 0,

                                                       template (omit) SRB_ToAddModList p_SRB_ToAddModList := omit,

                                                       template (omit) DRB_ToAddModList p_DRB_ToAddModList := omit,

                                                       template (omit) MAC_MainConfig_Type p_MAC_MainConfig := omit,

                                                       template (omit) PhysicalConfigDedicated p_PhysicalConfigDedicated := omit,

                                                       template (omit) EPS_QoSList_Type p_QoS := omit,

                                                       template (omit) Tft_List_Type p_Tft := omit,

                                                       boolean p_PDCP_StatusReportRequired := true,

                                                       template (omit) RLF_TimersAndConstants_r9 p_RLF_TimersAndConstants_r9 := omit,

                                                       template (omit) DownlinkAntennaGroupConfig_Type p_DownlinkAntennaGroupConfig := omit,

                                                       template (omit) DRB_Identity p_DrbFor2ndPDN := omit) runs on EUTRA_PTC

  { /* @sic R5-130491, R5-121736, R5-121737 - support of MO, MT video call: new parameter p_SRB_ToAddModList; function renamed to f_EUTRA_RRCConnectionReconfiguration_Common sic@ */

    /* @sic R5-144797: new parameter p_DrbFor2ndPDN sic@ */

    var integer n, m, i, v_DRB_Id;

    var DRB_Identity v_DrbFor2ndPDN;

    var EPS_BearerIdentity v_EpsBearerId;

    var integer v_NoOfDefaultBearers := 1;  /* @sic R5-144797 sic@ */

    var integer v_NoOfQos := 0;

    var integer v_NoOfTft := 0;

    var integer v_NoOfAddDRBs := p_NoOfAddDRBs_AM + p_NoOfAddDRBs_UM;  /* number of bearers to be configured additionally (i.e. apart from the default bearer)

                                                                          if v_NoOfAddDRBs=0 => only the default DRB is established */

    var template (value) EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_Info := f_EUTRA_CellInfo_GetFDD_TDD_Info (p_CellId); //Get FDD or TDD mode from Cell-configuration

    var template (value) DRB_ToAddModList v_DRB_ToAddModList;

    var template (value) MAC_MainConfig_Type v_MAC_MainConfig;

    var template (omit)  PhysicalConfigDedicated v_PhysicalConfigDedicated;

    var template (omit)  NAS_MSG_RequestList_Type v_NAS_MsgList;

    var template (value) RadioResourceConfigDedicated  v_RadioResourceConfigDedicated;

    var template (value) DRX_Config v_DRX_Config;

    var template (value) PhysicalConfigDedicated_AntennaInfo_Type v_AntennaInfo := f_EUTRA_CellInfo_GetAntennaInfoDedicated (p_CellId); //@sic R5-100786 sic@

    var template (value) TimingInfo_Type v_TimingInfo := cs_TimingInfo_Now;   /* @sic R5-104796 DRX support sic@ */

    var SubFrameTiming_Type v_SubFrameTiming;

    if (isvalue(p_DrbFor2ndPDN)) { v_NoOfDefaultBearers := v_NoOfDefaultBearers + 1; }  /* @sic R5-144797 sic@ */

    // -----------------------------------------------------

    // Check parameters are valid

    if (isvalue(p_QoS)) {

      v_NoOfQos := lengthof(p_QoS);

    }

    if (isvalue(p_Tft)) {

      v_NoOfTft := lengthof(p_Tft);

    }

    if (v_NoOfQos != v_NoOfAddDRBs) {

      FatalError (__FILE__, __LINE__, "invalid no. of QoS");

    }

    if (v_NoOfTft != v_NoOfAddDRBs) {

      FatalError (__FILE__, __LINE__, "invalid no. of TFT");

    }

    // -----------------------------------------------------

    // initialise v_DRB_ToAddModList

    if (isvalue(p_DRB_ToAddModList)) {                  // explicit setting for DRB_ToAddMod

      if (lengthof(p_DRB_ToAddModList) != v_NoOfAddDRBs + v_NoOfDefaultBearers) {

        FatalError (__FILE__, __LINE__, "invalid no. of DRB_ToAddMod");

      }

      v_DRB_ToAddModList := valueof(p_DRB_ToAddModList);

    } else {                                            // use default configuration

      v_DRB_Id := tsc_DRB1;                             // default bearer

      for (n := 0; n < (p_NoOfAddDRBs_AM + 1); n := n + 1) {                // default bearer included in the list (=> p_NoOfAddDRBs_AM + 1)

        if (p_PDCP_StatusReportRequired == true) {

          v_DRB_ToAddModList[n] := cs_508_DRB_ToAddMod_DEFAULT_AM(v_DRB_Id);

        }

        else {

          v_DRB_ToAddModList[n] := cs_DRB_ToAddMod_NoPDCP_StatusReport(v_DRB_Id);

        }

        v_DRB_Id := v_DRB_Id + 1;

      }

      for (m := 0; m < p_NoOfAddDRBs_UM; m := m + 1) {

        v_DRB_ToAddModList[n+m] := cs_508_DRB_ToAddMod_DEFAULT_UM (v_DRB_Id);

        v_DRB_Id := v_DRB_Id + 1;

      }

    }

    if (isvalue(p_DrbFor2ndPDN)) {  /* @sic R5-144797 sic@ */

      v_DrbFor2ndPDN := valueof(p_DrbFor2ndPDN);

      if (v_DrbFor2ndPDN < v_DRB_Id) {

        FatalError (__FILE__, __LINE__, "2nd default bearer is within the range of dedicated bearers");

      }

      v_DRB_ToAddModList[v_NoOfAddDRBs+1] := cs_508_DRB_ToAddMod_DEFAULT_AM(v_DrbFor2ndPDN);

    }

    // -----------------------------------------------------

    // initialise RRCConnectionReconfiguration

    if (isvalue(p_MAC_MainConfig)) {

      v_MAC_MainConfig := valueof(p_MAC_MainConfig);

    }

    else {

      if (p_NoOfAddDRBs_UM > 0) {                /* @sic R5-104796 DRX support sic@ */

        v_MAC_MainConfig := cs_508_MAC_MainConfig_Explicit_RBC_DrxS; // short DRX cycle acc. to note in 36.508 table 4.6.3-16

      }

      else {

        v_MAC_MainConfig := cs_508_MAC_MainConfig_Explicit_RBC_DrxL; // long DRX cycle acc. to note in 36.508 table 4.6.3-16

      }

    }

    if (isvalue(p_PhysicalConfigDedicated)) {

      v_PhysicalConfigDedicated := valueof(p_PhysicalConfigDedicated);

    } else {

      if (isvalue(p_SRB_ToAddModList)) {   // @sic R5-130491, R5-121736, R5-121737 - support of MO, MT video call: per default PhysicalConfigDedicated are needed only if SRB2 is configured too sic@

        v_PhysicalConfigDedicated := cs_508_PhysicalConfigDedicated_Default_RBC(v_EUTRA_FDD_TDD_Info.cqi_ReportConfig,

                                                                                v_EUTRA_FDD_TDD_Info.soundingRS_UL_Config,

                                                                                v_AntennaInfo);
      } else {

        v_PhysicalConfigDedicated := omit;

      }

    }

    v_RadioResourceConfigDedicated := cs_RadioResourceConfigDedicated_SrbDrbConfig(p_SRB_ToAddModList,           // @sic R5-130491, R5-121736, R5-121737 - support of MO, MT video call sic@

                                                                                   v_DRB_ToAddModList,

                                                                                   v_MAC_MainConfig,

                                                                                   v_PhysicalConfigDedicated,

                                                                                   p_RLF_TimersAndConstants_r9); // @sic R5s120808 sic@

    // -----------------------------------------------------

    // initialise NAS messages

    if (v_NoOfAddDRBs == 0) {

      v_NAS_MsgList := omit;                          // no NAS messages to be sent

    } else {

      for (i := 0; i < v_NoOfAddDRBs; i := i+1) {     // only for the additional DRBs (there is no ActDedEPSBearerCxtReq for the already existing default bearer)

        // get EpsBearerId from the corresponding DRB_ToAddModList; DRB_ToAddModList[0] is the default bearer which needs to be skipped

        v_EpsBearerId := int2hex(valueof(v_DRB_ToAddModList[i+1].eps_BearerIdentity), 1);  // i+1 to skip default DRB; valueof cannot be avoided here

        v_NAS_MsgList[i] := cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                           cs_ActDedEPSBearerCxtReq_QoS(v_EpsBearerId, valueof(p_QoS[i]), valueof(p_Tft[i])));    // valueof to avoid compiler warnings: neither p_QoS nor p_Tft can be omit as v_NoOfAddDRBs > 0

      }

    }

    if (p_NoOfAddDRBs_UM > 0) {    /* @sic R5-104796 DRX support sic@ */

      if (ispresent(v_MAC_MainConfig.explicitValue.drx_Config)) {                /* @sic R5120095 sic@ */

        v_DRX_Config := valueof(v_MAC_MainConfig.explicitValue.drx_Config);      /* valueof is ok as v_MAC_MainConfig.explicitValue.drx_Config is checked with 'ispresent' beforehand */

        f_EUTRA_SS_DrxCtrlConfig(p_CellId, cs_DrxCtrl_Config(v_DRX_Config));     /* configure DRX at the SS; @sic R5s100630 R5s120054 sic@ */

        v_SubFrameTiming := f_EUTRA_GetNextSendOccasion (p_CellId, 100 + ((v_NoOfAddDRBs - 1) * 20));  /* @sic R5-115770 change 5 sic@ */

        v_TimingInfo := cs_TimingInfo(v_SubFrameTiming.SFN.Number, v_SubFrameTiming.Subframe.Number);

      }

    }

    else {

      /* do nothing; for the time being we don't need any DRX configuration at the SS for DRX_L */

    }

    /* Step 8 */

    SRB.send(cas_SRB1_RrcNasPduList_REQ(p_CellId,

                                        v_TimingInfo,

                                        cs_RRCConnectionReconfiguration_Common(p_RRC_TI,

                                                                               omit,

                                                                               omit,

                                                                               v_RadioResourceConfigDedicated,

                                                                               omit),

                                        v_NAS_MsgList));

    if (ischosen(v_AntennaInfo.explicitValue)) {

      f_EUTRA_SS_CommonCellConfig(p_CellId, cas_PhysicalLayerConfigDL_CcchDcchDtchDL_Config_REQ(p_CellId,

                                                                                                cs_TimingInfo_Now,

                                                                                                cs_PhysicalLayerConfigDL_AntennaInfo(p_DownlinkAntennaGroupConfig),

                                                                                                cs_DciDLInfo_Auto(cs_DciInfo_CcchDcchDtchDL_Dci1A(f_EUTRA_CellInfo_GetDL_ChBandwidth(p_CellId))),

                                                                                                cs_AntennaInfoDedicated_r8andLater(v_AntennaInfo.explicitValue)))

    }

  }

After change

        function f_EUTRA_RRCConnectionReconfiguration_Common(EUTRA_CellId_Type p_CellId,

                                                       RRC_TransactionIdentifier p_RRC_TI,

                                                       UInt_Type p_NoOfAddDRBs_AM := 0,

                                                       UInt_Type p_NoOfAddDRBs_UM := 0,

                                                       template (omit) SRB_ToAddModList p_SRB_ToAddModList := omit,

                                                       template (omit) DRB_ToAddModList p_DRB_ToAddModList := omit,

                                                       template (omit) MAC_MainConfig_Type p_MAC_MainConfig := omit,

                                                       template (omit) PhysicalConfigDedicated p_PhysicalConfigDedicated := omit,

                                                       template (omit) EPS_QoSList_Type p_QoS := omit,

                                                       template (omit) Tft_List_Type p_Tft := omit,

                                                       boolean p_PDCP_StatusReportRequired := true,

                                                       template (omit) RLF_TimersAndConstants_r9 p_RLF_TimersAndConstants_r9 := omit,

                                                       template (omit) DownlinkAntennaGroupConfig_Type p_DownlinkAntennaGroupConfig := omit,

                                                       template (omit) DRB_Identity p_DrbFor2ndPDN := omit) runs on EUTRA_PTC

  { /* @sic R5-130491, R5-121736, R5-121737 - support of MO, MT video call: new parameter p_SRB_ToAddModList; function renamed to f_EUTRA_RRCConnectionReconfiguration_Common sic@ */

    /* @sic R5-144797: new parameter p_DrbFor2ndPDN sic@ */

    var integer n, m, i, v_DRB_Id;

    var DRB_Identity v_DrbFor2ndPDN;

    var EPS_BearerIdentity v_EpsBearerId;

    var integer v_NoOfDefaultBearers := 1;  /* @sic R5-144797 sic@ */

    var integer v_NoOfQos := 0;

    var integer v_NoOfTft := 0;

    var integer v_NoOfAddDRBs := p_NoOfAddDRBs_AM + p_NoOfAddDRBs_UM;  /* number of bearers to be configured additionally (i.e. apart from the default bearer)

                                                                          if v_NoOfAddDRBs=0 => only the default DRB is established */

    var template (value) EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_Info := f_EUTRA_CellInfo_GetFDD_TDD_Info (p_CellId); //Get FDD or TDD mode from Cell-configuration

    var template (value) DRB_ToAddModList v_DRB_ToAddModList;

    var template (value) MAC_MainConfig_Type v_MAC_MainConfig;

    var template (omit)  PhysicalConfigDedicated v_PhysicalConfigDedicated;

    var template (omit)  NAS_MSG_RequestList_Type v_NAS_MsgList;

    var template (value) RadioResourceConfigDedicated  v_RadioResourceConfigDedicated;

    var template (value) DRX_Config v_DRX_Config;

    var template (value) PhysicalConfigDedicated_AntennaInfo_Type v_AntennaInfo := f_EUTRA_CellInfo_GetAntennaInfoDedicated (p_CellId); //@sic R5-100786 sic@

    var template (value) TimingInfo_Type v_TimingInfo := cs_TimingInfo_Now;   /* @sic R5-104796 DRX support sic@ */

    var SubFrameTiming_Type v_SubFrameTiming;

    if (isvalue(p_DrbFor2ndPDN)) { v_NoOfDefaultBearers := v_NoOfDefaultBearers + 1; }  /* @sic R5-144797 sic@ */

    // -----------------------------------------------------

    // Check parameters are valid

    if (isvalue(p_QoS)) {

      v_NoOfQos := lengthof(p_QoS);

    }

    if (isvalue(p_Tft)) {

      v_NoOfTft := lengthof(p_Tft);

    }

    if (v_NoOfQos != v_NoOfAddDRBs) {

      FatalError (__FILE__, __LINE__, "invalid no. of QoS");

    }

    if (v_NoOfTft != v_NoOfAddDRBs) {

      FatalError (__FILE__, __LINE__, "invalid no. of TFT");

    }

    // -----------------------------------------------------

    // initialise v_DRB_ToAddModList

    if (isvalue(p_DRB_ToAddModList)) {                  // explicit setting for DRB_ToAddMod

      if (lengthof(p_DRB_ToAddModList) != v_NoOfAddDRBs + v_NoOfDefaultBearers) {

        FatalError (__FILE__, __LINE__, "invalid no. of DRB_ToAddMod");

      }

      v_DRB_ToAddModList := valueof(p_DRB_ToAddModList);

    } else {                                            // use default configuration

      v_DRB_Id := tsc_DRB1;                             // default bearer

      for (n := 0; n < (p_NoOfAddDRBs_AM + 1); n := n + 1) {                // default bearer included in the list (=> p_NoOfAddDRBs_AM + 1)

        if (p_PDCP_StatusReportRequired == true) {

          v_DRB_ToAddModList[n] := cs_508_DRB_ToAddMod_DEFAULT_AM(v_DRB_Id);

        }

        else {

          v_DRB_ToAddModList[n] := cs_DRB_ToAddMod_NoPDCP_StatusReport(v_DRB_Id);

        }

        v_DRB_Id := v_DRB_Id + 1;

      }

      for (m := 0; m < p_NoOfAddDRBs_UM; m := m + 1) {

        v_DRB_ToAddModList[n+m] := cs_508_DRB_ToAddMod_DEFAULT_UM (v_DRB_Id);

        v_DRB_Id := v_DRB_Id + 1;

      }

    }

    if (isvalue(p_DrbFor2ndPDN)) {  /* @sic R5-144797 sic@ */

      v_DrbFor2ndPDN := valueof(p_DrbFor2ndPDN);

      if (v_DrbFor2ndPDN < v_DRB_Id) {

        FatalError (__FILE__, __LINE__, "2nd default bearer is within the range of dedicated bearers");

      }

      v_DRB_ToAddModList[v_NoOfAddDRBs+1] := cs_508_DRB_ToAddMod_DEFAULT_AM(v_DrbFor2ndPDN);

    }

    // -----------------------------------------------------

    // initialise RRCConnectionReconfiguration

    if (isvalue(p_MAC_MainConfig)) {

      v_MAC_MainConfig := valueof(p_MAC_MainConfig);

    }

    else {

      if (p_NoOfAddDRBs_UM > 0) {                /* @sic R5-104796 DRX support sic@ */

        v_MAC_MainConfig := cs_508_MAC_MainConfig_Explicit_RBC_DrxS; // short DRX cycle acc. to note in 36.508 table 4.6.3-16

      }

      else {

        v_MAC_MainConfig := cs_508_MAC_MainConfig_Explicit_RBC_DrxL; // long DRX cycle acc. to note in 36.508 table 4.6.3-16

      }

    }

    if (isvalue(p_PhysicalConfigDedicated)) {

      v_PhysicalConfigDedicated := valueof(p_PhysicalConfigDedicated);

    } else {

      if (isvalue(p_SRB_ToAddModList)) {   // @sic R5-130491, R5-121736, R5-121737 - support of MO, MT video call: per default PhysicalConfigDedicated are needed only if SRB2 is configured too sic@

      //Anite@LCMTC

       if (pc_FDD_TypeB_HalfDuplex)

       {

            v_PhysicalConfigDedicated := cs_508_PhysicalConfigDedicated_Default_RBC_TypeB_HalfDuplex(

                                                                                v_AntennaInfo);

       }

       else

       {
        v_PhysicalConfigDedicated := cs_508_PhysicalConfigDedicated_Default_RBC(v_EUTRA_FDD_TDD_Info.cqi_ReportConfig,

                                                                                v_EUTRA_FDD_TDD_Info.soundingRS_UL_Config,

                                                                                v_AntennaInfo);

        }
      } else {

        v_PhysicalConfigDedicated := omit;

      }

    }

    v_RadioResourceConfigDedicated := cs_RadioResourceConfigDedicated_SrbDrbConfig(p_SRB_ToAddModList,           // @sic R5-130491, R5-121736, R5-121737 - support of MO, MT video call sic@

                                                                                   v_DRB_ToAddModList,

                                                                                   v_MAC_MainConfig,

                                                                                   v_PhysicalConfigDedicated,

                                                                                   p_RLF_TimersAndConstants_r9); // @sic R5s120808 sic@

    // -----------------------------------------------------

    // initialise NAS messages

    if (v_NoOfAddDRBs == 0) {

      v_NAS_MsgList := omit;                          // no NAS messages to be sent

    } else {

      for (i := 0; i < v_NoOfAddDRBs; i := i+1) {     // only for the additional DRBs (there is no ActDedEPSBearerCxtReq for the already existing default bearer)

        // get EpsBearerId from the corresponding DRB_ToAddModList; DRB_ToAddModList[0] is the default bearer which needs to be skipped

        v_EpsBearerId := int2hex(valueof(v_DRB_ToAddModList[i+1].eps_BearerIdentity), 1);  // i+1 to skip default DRB; valueof cannot be avoided here

        v_NAS_MsgList[i] := cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                           cs_ActDedEPSBearerCxtReq_QoS(v_EpsBearerId, valueof(p_QoS[i]), valueof(p_Tft[i])));    // valueof to avoid compiler warnings: neither p_QoS nor p_Tft can be omit as v_NoOfAddDRBs > 0

      }

    }

    if (p_NoOfAddDRBs_UM > 0) {    /* @sic R5-104796 DRX support sic@ */

      if (ispresent(v_MAC_MainConfig.explicitValue.drx_Config)) {                /* @sic R5120095 sic@ */

        v_DRX_Config := valueof(v_MAC_MainConfig.explicitValue.drx_Config);      /* valueof is ok as v_MAC_MainConfig.explicitValue.drx_Config is checked with 'ispresent' beforehand */

        f_EUTRA_SS_DrxCtrlConfig(p_CellId, cs_DrxCtrl_Config(v_DRX_Config));     /* configure DRX at the SS; @sic R5s100630 R5s120054 sic@ */

        v_SubFrameTiming := f_EUTRA_GetNextSendOccasion (p_CellId, 100 + ((v_NoOfAddDRBs - 1) * 20));  /* @sic R5-115770 change 5 sic@ */

        v_TimingInfo := cs_TimingInfo(v_SubFrameTiming.SFN.Number, v_SubFrameTiming.Subframe.Number);

      }

    }

    else {

      /* do nothing; for the time being we don't need any DRX configuration at the SS for DRX_L */

    }

    /* Step 8 */

    SRB.send(cas_SRB1_RrcNasPduList_REQ(p_CellId,

                                        v_TimingInfo,

                                        cs_RRCConnectionReconfiguration_Common(p_RRC_TI,

                                                                               omit,

                                                                               omit,

                                                                               v_RadioResourceConfigDedicated,

                                                                               omit),

                                        v_NAS_MsgList));

    if (ischosen(v_AntennaInfo.explicitValue)) {

      f_EUTRA_SS_CommonCellConfig(p_CellId, cas_PhysicalLayerConfigDL_CcchDcchDtchDL_Config_REQ(p_CellId,

                                                                                                cs_TimingInfo_Now,

                                                                                                cs_PhysicalLayerConfigDL_AntennaInfo(p_DownlinkAntennaGroupConfig),

                                                                                                cs_DciDLInfo_Auto(cs_DciInfo_CcchDcchDtchDL_Dci1A(f_EUTRA_CellInfo_GetDL_ChBandwidth(p_CellId))),

                                                                                                cs_AntennaInfoDedicated_r8andLater(v_AntennaInfo.explicitValue)))

    }

  }

EUTRA_RRCSteps.ttcn
Before change

   function f_EUTRA_RRCConnectionReconfiguration_SRB2_1AM_DRB_Send(EUTRA_CellId_Type p_CellId,

                                                                  RRC_TransactionIdentifier p_RRC_TI := tsc_RRC_TI_Def,

                                                                  template (omit) NAS_MSG_Request_Type p_NAS_MSG_Request := omit,

                                                                  DRB_Identity p_DRB_Id := tsc_DRB1)

    runs on EUTRA_PTC

  { /* @sic R5-144797: parameter p_DRB_Id sic@  */

    var template (value) EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_CellInfo := f_EUTRA_CellInfo_GetFDD_TDD_Info(p_CellId);

    var template (value) PhysicalConfigDedicated_AntennaInfo_Type v_AntennaInfo := f_EUTRA_CellInfo_GetAntennaInfoDedicated (p_CellId);

    var template (omit)  NAS_MSG_RequestList_Type v_NasMsgList := omit;

    if (isvalue(p_NAS_MSG_Request)) {

      v_NasMsgList := {p_NAS_MSG_Request};

    }

        SRB.send(cas_SRB1_RrcNasPduList_REQ(p_CellId,

                                        cs_TimingInfo_Now,

                                        cs_RRCConnectionReconfiguration_SRB2_1AM_DRB_Est(p_RRC_TI,

                                                                                         v_EUTRA_FDD_TDD_CellInfo.cqi_ReportConfig,

                                                                                         v_EUTRA_FDD_TDD_CellInfo.soundingRS_UL_Config,

                                                                                         v_AntennaInfo,

                                                                                         p_DRB_Id),

                                        v_NasMsgList));
  }

After change

   function f_EUTRA_RRCConnectionReconfiguration_SRB2_1AM_DRB_Send(EUTRA_CellId_Type p_CellId,

                                                                  RRC_TransactionIdentifier p_RRC_TI := tsc_RRC_TI_Def,

                                                                  template (omit) NAS_MSG_Request_Type p_NAS_MSG_Request := omit,

                                                                  DRB_Identity p_DRB_Id := tsc_DRB1)

    runs on EUTRA_PTC

  { /* @sic R5-144797: parameter p_DRB_Id sic@  */

    var template (value) EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_CellInfo := f_EUTRA_CellInfo_GetFDD_TDD_Info(p_CellId);

    var template (value) PhysicalConfigDedicated_AntennaInfo_Type v_AntennaInfo := f_EUTRA_CellInfo_GetAntennaInfoDedicated (p_CellId);

    var template (omit)  NAS_MSG_RequestList_Type v_NasMsgList := omit;

    if (isvalue(p_NAS_MSG_Request)) {

      v_NasMsgList := {p_NAS_MSG_Request};

    }

    //Anite@LCMTC Change for Half Duplex

    if (pc_FDD_TypeB_HalfDuplex)

    {

        SRB.send(cas_SRB1_RrcNasPduList_REQ(p_CellId,

                                        cs_TimingInfo_Now,

                                        cs_RRCConnectionReconfiguration_SRB2_1AM_DRB_Est_TypeB_HalfDuplex(p_RRC_TI,

                                                                                         v_AntennaInfo,

                                                                                         p_DRB_Id),

                                        v_NasMsgList));

    }    

    else

    {
        SRB.send(cas_SRB1_RrcNasPduList_REQ(p_CellId,

                                        cs_TimingInfo_Now,

                                        cs_RRCConnectionReconfiguration_SRB2_1AM_DRB_Est(p_RRC_TI,

                                                                                         v_EUTRA_FDD_TDD_CellInfo.cqi_ReportConfig,

                                                                                         v_EUTRA_FDD_TDD_CellInfo.soundingRS_UL_Config,

                                                                                         v_AntennaInfo,

                                                                                         p_DRB_Id),

                                        v_NasMsgList));

    }
  }

After change (New Templates):
  //Anite@LCMTC: Start New Template

   template (value) PhysicalConfigDedicated cs_508_PhysicalConfigDedicated_Default_RBC_HO_TypeB_HalfDuplex(template (omit) PUCCH_ConfigDedicated p_PUCCH_ConfigDedicated := omit,

                                                                                         template (value) PhysicalConfigDedicated_AntennaInfo_Type p_AntennaInfo,  // @sic R5-100786 sic@

                                                                                         template (omit) SchedulingRequestConfig p_SchedulingRequestConfig := omit) :=

  /* Default values according to 36.508 cl. 4.8.2.1.6 Table 4.8.2.1.6-1 for condition RBC or RBC-HO:

     (for RBC pucch-ConfigDedicated and schedulingRequestConfig are 'not present' i.e. p_PUCCH_ConfigDedicated and p_SchedulingRequestConfig shall be omit */

  /* @status    APPROVED (LTE, LTE_A_IRAT, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT) */

  /* @sic R5s100473: additional changes (new parameter p_SchedulingRequestConfig) sic@ */

  /* @sic R5s120663 change 6: new parameter p_PUCCH_ConfigDedicated sic@ */

  /* @sic R5s150362 - Additional change: cds_PhysicalConfigDedicated_Common sic@ */

    cds_PhysicalConfigDedicated_Common(-, p_PUCCH_ConfigDedicated, -, -, -, -, p_AntennaInfo, p_SchedulingRequestConfig);

  //Anite@LCMTC: End

  //Anite@LCMTC : New Template

   template (value) PhysicalConfigDedicated cs_508_PhysicalConfigDedicated_Default_RBC_TypeB_HalfDuplex(

                                                                                      template (value) PhysicalConfigDedicated_AntennaInfo_Type p_AntennaInfo) :=

  /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, POS) */

  /* @sic R5s120663 change 6 sic@ */

  /* @sic R5s150362 - Additional change: cds_PhysicalConfigDedicated_Common sic@ */

    cds_PhysicalConfigDedicated_Common(-, -, -, -, -, -, p_AntennaInfo);

  //Anite@LCMTC : New Template

   template (value) PhysicalConfigDedicated cds_508_PhysicalConfigDedicated_Default_SRB1_TypeB_HalfDuplex(template (value) PUCCH_ConfigDedicated p_PUCCH_ConfigDedicated,

                                                                                        template (omit)  UplinkPowerControlDedicated p_UplinkPowerControlDedicated := cs_508_UplinkPowerControlDedicated_Default,

                                                                                        SR_PUCCH_ResourceIndex_Type p_SR_PUCCH_ResourceIndex,                          // @sic R5s090312 sic@

                                                                                        template (value) PhysicalConfigDedicated_AntennaInfo_Type p_AntennaInfo,       // @sic R5-100786 sic@

                                                                                        SR_ConfigIndex_Type p_SR_ConfigIndex)                                          // @sic R5-102177 sic@

    modifies cs_PhysicalConfigDedicated_Omit :=

  { /* Default values according to 36.508 cl. 4.8.2.1.6 Table 4.8.2.1.6-1 */

    /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, POS) */

    /* @sic R5s150362 - Additional change: modifies cs_PhysicalConfigDedicated_Omit; new parameter p_UplinkPowerControlDedicated sic@ */

    pdsch_ConfigDedicated := cs_508_PDSCH_ConfigDedicated_Default,

    pucch_ConfigDedicated := p_PUCCH_ConfigDedicated,

    pusch_ConfigDedicated := cs_508_PUSCH_ConfigDedicated_Default,

    uplinkPowerControlDedicated := p_UplinkPowerControlDedicated,

    cqi_ReportConfig := omit,   //Anite@LCMTC

    soundingRS_UL_ConfigDedicated := omit,   //Anite@LCMTC

    antennaInfo := p_AntennaInfo,                                                  // @sic R5-100786 sic@

    schedulingRequestConfig := cs_508_SchedulingRequest_Config_Default(p_SR_PUCCH_ResourceIndex, p_SR_ConfigIndex)

  };

  //Anite@LCMTC : End
  //Anite@LCMTC : New Template

   template (value) RadioResourceConfigDedicated cs_SRB2_1AM_DRB_Config_Def_TypeB_HalfDuplex( template (value) PhysicalConfigDedicated_AntennaInfo_Type p_AntennaInfo,

                                                                           DRB_Identity p_DRB_Id := tsc_DRB1) := // @sic R5-100786 sic@

    /* Values according to 36.508 cl. 4.6.3 Table 4.6.3-16 (using condition n=1 & m=0) */

    /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, POS) */

    /* @sic R5-144797: parameter p_DRB_Id sic@  */

    cs_508_RadioResourceConfigDedicated_SRB2({ cs_508_DRB_ToAddMod_DEFAULT_AM(p_DRB_Id) },  /* @sic R5s141149 change 1 sic@ */

                                             cs_508_MAC_MainConfig_Explicit_RBC_DrxL,

                                             cs_508_PhysicalConfigDedicated_Default_RBC_TypeB_HalfDuplex(p_AntennaInfo));  //@sic R5-100786 sic@

  //Anite@LCMTC: End

  //Anite@LCMTC: New Template

   template (value) DL_CCCH_Message cs_508_RRCConnectionSetup_TypeB_HalfDuplex(RRC_TransactionIdentifier p_RRC_TI,

                                                             template (value) PUCCH_ConfigDedicated p_PUCCH_ConfigDedicated,

                                                             SR_PUCCH_ResourceIndex_Type p_SR_PUCCH_ResourceIndex,                          // @sic R5s090312 sic@

                                                             template (value) PhysicalConfigDedicated_AntennaInfo_Type p_AntennaInfo,       // @sic R5-100786 sic@

                                                             SR_ConfigIndex_Type p_SR_ConfigIndex,                                          // @sic R5-102177 sic@

                                                             TimeAlignmentTimer p_TimeAlignmentTimer := sf750) :=                           // @sic R5s110469 sic@

  { /* Default values according to 36.508 cl. 4.6.1 Table 4.6.1-17 */

    /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, POS) */

    message_ := {

      c1 := {

        rrcConnectionSetup := {

          rrc_TransactionIdentifier := p_RRC_TI,

          criticalExtensions := {

            c1 := {

              rrcConnectionSetup_r8 := {

                radioResourceConfigDedicated :=

                cs_508_RadioResourceConfigDedicated_SRB1(cs_MAC_MainConfig_SRB1(p_TimeAlignmentTimer),

                                                         cds_508_PhysicalConfigDedicated_Default_SRB1_TypeB_HalfDuplex(p_PUCCH_ConfigDedicated,

                                                                                                      -,

                                                                                                      p_SR_PUCCH_ResourceIndex,

                                                                                                      p_AntennaInfo,

                                                                                                      p_SR_ConfigIndex)),

                nonCriticalExtension := omit

              }

            }

          }

        }

      }

    }

  };

  //Anite@LCMTC: End

  //Anite@LCMTC: New Template

   template (value) DL_DCCH_Message cs_RRCConnectionReconfiguration_SRB2_1AM_DRB_Est_TypeB_HalfDuplex(RRC_TransactionIdentifier p_RRC_TI,

                                                                                    template (value) PhysicalConfigDedicated_AntennaInfo_Type p_AntennaInfo,   // @sic R5-100786 sic@

                                                                                    DRB_Identity p_DRB_Id := tsc_DRB1) :=

    /* Values according to 36.508 cl. 4.6.1 Table 4.6.1-8 (using condition SRB2-DRB(n, m), where n=1 & m=0) */

    /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, POS) */

    /* @sic R5-144797: parameter p_DRB_Id sic@  */

    cs_RRCConnectionReconfiguration_Common(p_RRC_TI,

                                           omit,

                                           omit,

                                           cs_SRB2_1AM_DRB_Config_Def_TypeB_HalfDuplex(p_AntennaInfo, p_DRB_Id),

                                           omit);

  //Anite@LCMTC: End
Change 2 – Correction to Function “f_EUTRA_RRC_ConnectionRelease_Common”
	Function name
	f_EUTRA_RRC_ConnectionRelease_Common

	Reason for change
	The current test model to test Low Cost MTC test cases in half duplex operation restricts UE UL transmission to subframe #2 only and DL reception to subframe #8 only. However, in function f_EUTRA_RRC_ConnectionRelease_Common, it is not ensured that subframe 8 is selected for transmission of RRCConnectionRelease message. 

	Summary of change
	Hardcoded the value of subframe to be used for message transmission based on the UE capability to support Half Duplex operation, pc_FDD_TypeB_HalfDuplex (Note) is used for the same.
Note: Please note that this checking should be based on a pixit parameter. So probably, a prose CR is required for adding a nex pixit parameter for testing in Half Duplex operation and that should be used to enforce half duplex changes. We can provide a draft if it is agreed.

	TTCN module
	EUTRA\EUTRA_RRCSteps.ttcn

	MCC160 Comment
	


Before change

     function f_EUTRA_RRC_ConnectionRelease_Common(EUTRA_CellId_Type p_CellId,

                                                template (value) DL_DCCH_Message p_RRCConnectionRelease,

                                                template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now) runs on EUTRA_PTC

  { /* @sic R5-110803 change 13 sic@ */

    /* @sic R5s110634 MCC160 implementation of change 1.1: new parameter p_DrbConfigList sic@ */

    /* @sic R5-123740 change 2: additional delay of 640ms at the end of the generic RRC Connection Release procedure (result of AP#55.04, replacing R5-1421779 change 14) sic@ */

    /* @sic R5s150338: p_DrbConfigList removed sic@ */

    var SubFrameTiming_Type v_TimingAtT;   // time T: sending of RRCConnectionRelease acc. to 36.523-3 clause 7.18

    var SubFrameTiming_Type v_TimingNow;

    var integer v_MilliSecondsDelay := tsc_EUTRA_DelayBeforeIntraCellHO;

    var integer v_Duration;

    var integer v_SubframeNumber; //Anite@LCMTC

    timer t_Timer;

    if (f_TimingInfo_IsNow(p_TimingInfo)) {

      v_TimingAtT := f_EUTRA_GetNextSendOccasion(p_CellId, v_MilliSecondsDelay);

      v_Duration := v_MilliSecondsDelay;

    }

    else {

      // dedicated timing info is given

      v_TimingAtT := valueof(p_TimingInfo.SubFrame);

      v_TimingNow := f_EUTRA_GetCurrentTiming(p_CellId);

      v_Duration := f_EUTRA_SubFrameTimingDuration(v_TimingNow, v_TimingAtT);

    }

    v_Duration := v_Duration + tsc_EUTRA_DelayAfterRRCConnectionRelease;   // @sic R5s120683 sic@

    t_Timer.start (int2float(v_Duration) / 1000.0);

    SRB.send(cas_SRB1_RrcPdu_REQ(p_CellId,

                                 cs_TimingInfo(v_TimingAtT.SFN.Number, v_TimingAtT.Subframe.Number), //Anite@LCMTC

                                 p_RRCConnectionRelease));

    f_EUTRA_RRC_ConnectionRelease_Local(p_CellId, v_TimingAtT);

    alt {

      [] t_Timer.timeout {}

      [] SRB.check(receive(car_SRB0_RrcPdu_IND(?, cr_RRCConnectionRequest(cr_EstablishmentCause_Any)))) {

        // there is an RRCConnectionRequest sent by the UE which can be received by TTCN-3 in a sub-sequent function

        t_Timer.stop;

      }

    }

  }
After change

     function f_EUTRA_RRC_ConnectionRelease_Common(EUTRA_CellId_Type p_CellId,

                                                template (value) DL_DCCH_Message p_RRCConnectionRelease,

                                                template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now) runs on EUTRA_PTC

  { /* @sic R5-110803 change 13 sic@ */

    /* @sic R5s110634 MCC160 implementation of change 1.1: new parameter p_DrbConfigList sic@ */

    /* @sic R5-123740 change 2: additional delay of 640ms at the end of the generic RRC Connection Release procedure (result of AP#55.04, replacing R5-1421779 change 14) sic@ */

    /* @sic R5s150338: p_DrbConfigList removed sic@ */

    var SubFrameTiming_Type v_TimingAtT;   // time T: sending of RRCConnectionRelease acc. to 36.523-3 clause 7.18

    var SubFrameTiming_Type v_TimingNow;

    var integer v_MilliSecondsDelay := tsc_EUTRA_DelayBeforeIntraCellHO;

    var integer v_Duration;

    var integer v_SubframeNumber; //Anite@LCMTC
    timer t_Timer;

    if (f_TimingInfo_IsNow(p_TimingInfo)) {

      v_TimingAtT := f_EUTRA_GetNextSendOccasion(p_CellId, v_MilliSecondsDelay);

      v_Duration := v_MilliSecondsDelay;

    }

    else {

      // dedicated timing info is given

      v_TimingAtT := valueof(p_TimingInfo.SubFrame);

      v_TimingNow := f_EUTRA_GetCurrentTiming(p_CellId);

      v_Duration := f_EUTRA_SubFrameTimingDuration(v_TimingNow, v_TimingAtT);

    }

    if (pc_FDD_TypeB_HalfDuplex)

    {

     v_SubframeNumber := 8; //Anite@LCMTC

    }

    else

    {

     v_SubframeNumber := v_TimingAtT.Subframe.Number; //Anite@LCMTC     

    }
    v_Duration := v_Duration + tsc_EUTRA_DelayAfterRRCConnectionRelease;   // @sic R5s120683 sic@

    t_Timer.start (int2float(v_Duration) / 1000.0);

    SRB.send(cas_SRB1_RrcPdu_REQ(p_CellId,

                                 cs_TimingInfo(v_TimingAtT.SFN.Number, v_SubframeNumber /*v_TimingAtT.Subframe.Number*/), //Anite@LCMTC

                                 p_RRCConnectionRelease));

    f_EUTRA_RRC_ConnectionRelease_Local(p_CellId, v_TimingAtT);

    alt {

      [] t_Timer.timeout {}

      [] SRB.check(receive(car_SRB0_RrcPdu_IND(?, cr_RRCConnectionRequest(cr_EstablishmentCause_Any)))) {

        // there is an RRCConnectionRequest sent by the UE which can be received by TTCN-3 in a sub-sequent function

        t_Timer.stop;

      }

    }

  }
5 Execution Log Files

Sequans Calliope-Cat0
The Sequans Calliope-Cat0 UE passed this test case on Anite Conformance Toolset Solution on LTE FDD band 4.The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file: 
TC_8_5_4_4_Log.html
6 References
	[1]
	R5s160587: Supporting information for addition of LTE-A Rel-12 LC-MTC test case 8.5.4.4 in Half Duplex Mode. This archive comprises html and xml format execution log files and PICS/PIXIT settings file respectively


